



# Premiers pas en Maple (Partie 1 de 5)

© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

Pour voir le contenu des différentes sections, cliquer avec la souris sur le triangle ► précédant le titre. La section se déploiera ▼ et son contenu sera affiché.

(un clic gauche ou un clic droit sur le triangle permettra de rétracter la section et son contenu sera alors masqué).

Pour votre confort, vous pouvez ajuster la taille de l'affichage à l'aide de la commande *Facteur de zoom* du menu *Affichage*.

Bonne lecture à tous !

\* Ce document Maple est exécutable avec la version 2020.2

## ▼ Initialisation

```
[> restart;
```

## ▼ Objectif

Le but de cette première partie est de donner à l'élève un bon départ avec Maple. Cette première partie consiste donc à initier l'élève à l'environnement interactif du logiciel Maple, à le familiariser avec la syntaxe informatique, avec le mécanisme de la simplification automatique et à lui montrer la différence que tout utilisateur de Maple doit faire entre variable mathématique (variable libre) et variable informatique (variable affectée).

## ▼ Déroulement

Avant même d'exécuter les requêtes, l'élève doit faire d'abord une lecture attentive de la présentation des différentes transpositions en Maple des notions mathématiques. Pour profiter au maximum de la présentation des différents éléments fondamentaux présentés dans ce document, il ne faut surtout pas être un lecteur passif. Au contraire, au fil du déroulement de cette feuille Maple, l'élève devra, de plus en plus, anticiper le résultat qui sera affiché et de se convaincre, à chaque fois, que le résultat obtenu est le résultat attendu.

## ▼ Opérateurs logiques et fonctions mathématiques

Voir document intitulé «**Opérateurs\_et\_fonctions.pdf**». On retrouvera ce document à l'adresse <http://plantagne.ca/Complements-Maple>

## ▼ Maple en tant que « calculatrice »

À l'onglet *Précision* du menu Outils → Options... , les deux premières cases ne doivent pas être cochées. Cela va permettre d'avoir un contrôle sur le nombre de chiffres affichés pour les prochaines requêtes de cette

section. Plus loin dans ce document, nous reviendrons sur la variable `Digits` et le nombre de chiffres significatifs dans un résultat.

Les expressions arithmétiques sont simplifiées dans les rationnels  $\mathbb{Q}$  que si elles ne contiennent aucun nombre décimal.

```
> 1153*(3^58+5^40)/(1-7^36);
      - 7958626149683391438912504722021
      1325865422929826735889511690800
```

(5.1)

Si au moins un des nombres est un nombre décimal, le résultat sera automatiquement un nombre décimal.

```
> 1153*(3.^58+5^40)/(1-7^36);
      -6.002589718
```

(5.2)

Notons qu'il y a 10 chiffres dans le résultat précédent.

Maple opère symboliquement la simplification en accord avec le mécanisme de la simplification automatique.

```
> sqrt(3)+2*sin(Pi/3)+cos(Pi/3);
      2√3 + 1/2
```

(5.3)

La macro-commande `evalf` est utile pour obtenir **une approximation décimale d'une expression exacte**. Si on ne précise pas le nombre de chiffres désirés, le résultat sera donné avec le nombre de chiffres spécifié par la variable d'environnement `Digits`. La valeur par défaut de cette variable est 10.

Obtenons une approximation du nombre  $\frac{1153(3^{58} + 5^{40})}{1 - 7^{36}}$ .

```
> evalf(1153*(3^58+5^40)/(1-7^36));
      -6.002589714
```

(5.4)

Obtenons une autre approximation mais en précisant cette fois que l'on désire 50 chiffres. Il suffit de préciser la valeur 50 en option dans la macro-commande `evalf`. Cet ajout ne sert qu'à préciser ponctuellement le nombre de chiffres désirés sans pour autant modifier la valeur de la variable d'environnement `Digits`.

```
> evalf[50](1153*(3^58+5^40)/(1-7^36));
      -6.0025897138918093290168976024446663471132545940837
```

(5.5)

```
> evalf[50](1153*(3.^58+5^40)/(1-7^36));
      -6.002589718
```

(5.6)

Pourquoi le dernier résultat n'a-t-il pas lui aussi 50 chiffres ?

La présence du point décimal pour le chiffre 3 amène d'abord la simplification automatique donnant un nombre de 10 chiffres (la valeur de la variable d'environnement `Digits` est 10) puis, la requête d'approximation avec 50 chiffres d'un nombre qui en compte que 10 ne peut avoir pour résultat que ces 10 chiffres.

Obtenons une approximation de la constante mathématique  $\pi$  avec 6000 chiffres.

```
> evalf[6000](Pi);
3.14159265358979323846264338327950288419716939937510582097494459230781640628620899\
862803482534211706798214808651328230664709384460955058223172535940812848111745\
028410270193852110555964462294895493038196442881097566593344612847564823378678\
316527120190914564856692346034861045432664821339360726024914127372458700660631\
```

(5.7)

558817488152092096282925409171536436789259036001133053054882046652138414695194\  
151160943305727036575959195309218611738193261179310511854807446237996274956735\  
188575272489122793818301194912983367336244065664308602139494639522473719070217\  
986094370277053921717629317675238467481846766940513200056812714526356082778577\  
134275778960917363717872146844090122495343014654958537105079227968925892354201\  
995611212902196086403441815981362977477130996051870721134999999837297804995105\  
973173281609631859502445945534690830264252230825334468503526193118817101000313\  
783875288658753320838142061717766914730359825349042875546873115956286388235378\  
759375195778185778053217122680661300192787661119590921642019893809525720106548\  
586327886593615338182796823030195203530185296899577362259941389124972177528347\  
913151557485724245415069595082953311686172785588907509838175463746493931925506\  
040092770167113900984882401285836160356370766010471018194295559619894676783744\  
944825537977472684710404753464620804668425906949129331367702898915210475216205\  
696602405803815019351125338243003558764024749647326391419927260426992279678235\  
478163600934172164121992458631503028618297455570674983850549458858692699569092\  
721079750930295532116534498720275596023648066549911988183479775356636980742654\  
252786255181841757467289097777279380008164706001614524919217321721477235014144\  
197356854816136115735255213347574184946843852332390739414333454776241686251898\  
356948556209921922218427255025425688767179049460165346680498862723279178608578\  
438382796797668145410095388378636095068006422512520511739298489608412848862694\  
560424196528502221066118630674427862203919494504712371378696095636437191728746\  
776465757396241389086583264599581339047802759009946576407895126946839835259570\  
982582262052248940772671947826848260147699090264013639443745530506820349625245\  
174939965143142980919065925093722169646151570985838741059788595977297549893016\  
175392846813826868386894277415599185592524595395943104997252468084598727364469\  
584865383673622262609912460805124388439045124413654976278079771569143599770012\  
961608944169486855584840635342207222582848864815845602850601684273945226746767\  
889525213852254995466672782398645659611635488623057745649803559363456817432411\  
251507606947945109659609402522887971089314566913686722874894056010150330861792\  
868092087476091782493858900971490967598526136554978189312978482168299894872265\  
880485756401427047755513237964145152374623436454285844479526586782105114135473\  
573952311342716610213596953623144295248493718711014576540359027993440374200731\  
057853906219838744780847848968332144571386875194350643021845319104848100537061\  
468067491927819119793995206141966342875444064374512371819217999839101591956181\  
467514269123974894090718649423196156794520809514655022523160388193014209376213\  
785595663893778708303906979207734672218256259966150142150306803844773454920260

541466592520149744285073251866600213243408819071048633173464965145390579626856\  
100550810665879699816357473638405257145910289706414011097120628043903975951567\  
715770042033786993600723055876317635942187312514712053292819182618612586732157\  
919841484882916447060957527069572209175671167229109816909152801735067127485832\  
228718352093539657251210835791513698820914442100675103346711031412671113699086\  
585163983150197016515116851714376576183515565088490998985998238734552833163550\  
764791853589322618548963213293308985706420467525907091548141654985946163718027\  
098199430992448895757128289059232332609729971208443357326548938239119325974636\  
673058360414281388303203824903758985243744170291327656180937734440307074692112\  
019130203303801976211011004492932151608424448596376698389522868478312355265821\  
314495768572624334418930396864262434107732269780280731891544110104468232527162\  
010526522721116603966655730925471105578537634668206531098965269186205647693125\  
705863566201855810072936065987648611791045334885034611365768675324944166803962\  
657978771855608455296541266540853061434443185867697514566140680070023787765913\  
440171274947042056223053899456131407112700040785473326993908145466464588079727\  
082668306343285878569830523580893306575740679545716377525420211495576158140025\  
012622859413021647155097925923099079654737612551765675135751782966645477917450\  
112996148903046399471329621073404375189573596145890193897131117904297828564750\  
320319869151402870808599048010941214722131794764777262241425485454033215718530\  
614228813758504306332175182979866223717215916077166925474873898665494945011465\  
406284336639379003976926567214638530673609657120918076383271664162748888007869\  
256029022847210403172118608204190004229661711963779213375751149595015660496318\  
629472654736425230817703675159067350235072835405670403867435136222247715891504\  
953098444893330963408780769325993978054193414473774418426312986080998886874132\  
604721569516239658645730216315981931951673538129741677294786724229246543668009\  
806769282382806899640048243540370141631496589794092432378969070697794223625082\  
216889573837986230015937764716512289357860158816175578297352334460428151262720\  
373431465319777741603199066554187639792933441952154134189948544473456738316249\  
934191318148092777710386387734317720754565453220777092120190516609628049092636\  
019759882816133231666365286193266863360627356763035447762803504507772355471058\  
595487027908143562401451718062464362679456127531813407833033625423278394497538\  
243720583531147711992606381334677687969597030983391307710987040859133746414428\  
227726346594704745878477872019277152807317679077071572134447306057007334924369\  
311383504931631284042512192565179806941135280131470130478164378851852909285452\  
011658393419656213491434159562586586557055269049652098580338507224264829397285\  
84783163057775606888764462482468579260395352773480304802900587607582510474709

164396136267604492562742042083208566119062545433721315359584506877246

Ce résultat est seulement pour épater la galerie. À part l'intérêt qu'on peut avoir quant à la manière dont Maple s'y prend pour donner toutes ces décimales de  $\pi$ , rares seront les occasions de faire des calculs avec un tel nombre de décimales. Peut-on faire afficher plus de décimales de  $\pi$ , bien sûr mais avec un temps de calcul nécessairement plus long.

Par contre, dans la plupart des calculs, 10 chiffres dans les nombres suffiront mais il faudra être vigilant afin de tenir compte de la propagation des erreurs d'arrondis dans les calculs intermédiaires. Nous allons en discuter dans l'avant-dernière section de ce document.

```
> evalf[15](1/2);  
0.5000000000000000 (5.8)
```

On constate dans le résultat de la requête précédente, que les zéros après le chiffre 5 ne sont pas des chiffres significatifs. Voici comment procéder afin de ne pas faire afficher les zéros non significatifs.

```
> interface(typesetting=extended); # Pour s'assurer le niveau de  
composition étendue  
Typesetting:-Settings(striptrailing=true);  
extended  
false (5.9)
```

```
> evalf[15](1/2);  
0.5 (5.10)
```

Les opérateurs systèmes [%](#), [%%](#) et [%%%](#) servent à pointer vers une expression déjà calculée pour en refaire le calcul:

`%` est pour le dernier résultat calculé (également avec l'opérateur `_ans`)

`%%` est pour le pénultième (*avant-dernier*) résultat calculé

`%%%` est pour l'antépénultième (*avant-avant-dernier*) résultat calculé

La constante mathématique  $e$  est saisi avec la macro-commande [exp](#).

```
> exp(1);  
e (5.11)
```

```
> evalf[20](%);  
2.7182818284590452354 (5.12)
```

```
> evalf[30](%);  
2.7182818284590452354 (5.13)
```

Pourquoi le dernier résultat n'est-il pas affiché avec 30 chiffres ? Que faut-il modifier dans la dernière requête pour qu'il en soit ainsi ?

Parce que la requête `evalf[30](%)`; pointe vers le calcul précédent, qui a consisté en une approximation de  $e$  avec 20 chiffres, d'où le résultat avec 20 chiffres. Voici plutôt la séquence des requêtes qu'il aurait fallu faire exécuter.

```
> exp(1);  
e (5.14)
```

```
> evalf[20](%);  
2.7182818284590452354 (5.15)
```

```
> evalf[30](%%);
```

$$2.71828182845904523536028747135 \quad (5.16)$$

L'argument des fonctions trigonométriques doivent être **spécifié en radians**.

```
> sin(13*Pi/4);
```

$$-\frac{\sqrt{2}}{2} \quad (5.17)$$

Si l'angle spécifié est en degrés, nous devons alors le convertir en radians en utilisant la macro-commande [convert](#).

```
> cos(225*degrees);
```

$$\cos(225 \text{ degrees}) \quad (5.18)$$

```
> cos(convert(225*degrees,radians));
```

$$-\frac{\sqrt{2}}{2} \quad (5.19)$$

```
> 225*degrees = convert(225*degrees,radians);
```

$$225 \text{ degrees} = \frac{5\pi}{4} \quad (5.20)$$

## Mécanisme de la simplification automatique

*Les nombres rationnels et les radicaux sont réduits automatiquement à leur plus simple expression.*

Soit les expressions  $\frac{123456}{987654}$  et  $\sqrt{63}$ .

```
> 123456/987654;
```

$$\frac{20576}{164609} \quad (6.1.1)$$

```
> sqrt(63);
```

$$3\sqrt{7} \quad (6.1.2)$$

Il est possible de saisir plus d'une requête à la fois dans une même zone de requêtes. C'est ce qui sera fait dans cette section-ci afin d'alléger la présentation. Par contre, dans le contexte de la résolution de problèmes, cela n'est pas recommandé. Le développement exigé par votre professeur sera comme celui qui est fait dans les développements manu scriptus: une ligne à la fois.

*Les opérations  $+$ ,  $\times$  et d'élevation à une puissance entières et rationnelles sont effectuées automatiquement. Le résultat est aussi automatiquement réduit à sa plus simple expression.*

Soit les expressions  $\frac{2}{3} + \frac{1}{4} - \frac{5}{12}$ ,  $\left(\frac{12}{13}\right)^5$  et  $\frac{3 \cdot 21}{7 \cdot 52}$ .

```
> 2/3+1/4-5/12;
(12/13)^5;
3/7*21/52;
```

$$\frac{1}{2}$$

$$\frac{248832}{371293}$$

$$\frac{9}{52} \quad (6.2.1)$$

Dans les expressions rationnelles, les facteurs syntaxiquement identiques au numérateur et au dénominateur sont automatiquement simplifiés, et ce sans évaluation (à la condition que l'expression ne soit pas la règle de calcul des images d'une fonction).

Soit les expressions  $\frac{2x}{x}$  et  $\frac{(x+1)^2 x^3}{x(x+1)^3}$ .

```
> 2*x/x;
(x+1)^2*x^3/(x*(x+1)^3);
```

$$\frac{x^2}{x+1} \quad (6.3.1)$$

```
> f:=x->(x+1)^2*x^3/(x*(x+1)^3);
```

$$f := x \mapsto \frac{(x+1)^2 \cdot x^3}{x \cdot (x+1)^3} \quad (6.3.2)$$

La réduction des termes semblables et les lois des exposants sont automatiquement appliqués

Soit les expressions  $a - b + a$  et  $(x+y)^2 \sqrt{x+y}$ .

```
> a-b+a;
(x+y)^2*sqrt(x+y);
(x^3)^2;
```

$$\frac{2a - b}{(x+y)^5 / 2} x^6 \quad (6.4.1)$$

La distributivité de la multiplication sur l'addition est automatique

Soit l'expression  $\frac{3(a+b)}{7}$ .

```
> 3/7*(a+b);
```

$$\frac{3a}{7} + \frac{3b}{7} \quad (6.5.1)$$

Le signe moins devant l'indéterminée dans la valeur absolue, dans les fonctions trigonométriques sinus et cosinus

et dans les fonctions trigonométriques réciproques est automatiquement simplifié

Soit les expressions  $|-x|$ ,  $\sin(-x)$ ,  $\cos(-x)$ ,  $\cot(-x)$ ,  $\arcsin(-x)$  et  $\arccos(-x)$ .

```
> abs(-x);
sin(-x);
cos(-x);
```

```
cot(-x);
arcsin(-x);
arccos(-x);
```

$$\begin{aligned} &|x| \\ &-\sin(x) \\ &\cos(x) \\ &-\cot(x) \\ &-\arcsin(x) \\ &\pi - \arccos(x) \end{aligned} \tag{6.6.1}$$

Les identités trigonométriques des angles complémentaires et des angles supplémentaires et autres angles sont utilisées pour réduire automatiquement les expressions trigonométriques

Soit les expressions  $\tan\left(x + \frac{\pi}{2}\right)$ ,  $\sin\left(x + \frac{\pi}{2}\right)$ ,  $\cos(x + \pi)$ ,  $\sin\left(-x - \frac{\pi}{4}\right)$  et  $\sin\left(-x + \frac{\pi}{5}\right)$ .

```
> tan(x+Pi/2);
sin(x+Pi/2);
cos(x+Pi);
sin(-x-Pi/4);
sin(-x+Pi/5);
```

$$\begin{aligned} &-\cot(x) \\ &\cos(x) \\ &-\cos(x) \\ &-\sin\left(x + \frac{\pi}{4}\right) \\ &\cos\left(x + \frac{3\pi}{10}\right) \end{aligned} \tag{6.7.1}$$

Tout calcul de logarithmes dans une base autre que la base  $e$  est automatiquement converti dans la base  $e$

```
> log[a](12);
```

$$\frac{\ln(12)}{\ln(a)} \tag{6.8.1}$$

De plus, l'expression  $e^{\ln(x)}$  est simplifiée automatiquement par  $x$ , mais l'expression  $\ln(e^x)$  ne l'est pas. La définition de la fonction logarithme dans les nombres complexes ne permet pas cette simplification automatique.

```
> exp(ln(x));
```

$$x \tag{6.8.2}$$

```
> ln(exp(x));
```

$$\ln(e^x) \tag{6.8.3}$$

Maple comporte une table d'affectation système qui lui permet de simplifier automatiquement plusieurs expressions trigonométriques, logarithmiques et d'autres expressions numériques remarquables

Soit les expressions numériques  $\cos\left(\frac{\pi}{6}\right)$ ,  $\tan\left(\frac{17\pi}{3}\right)$  et  $\ln(1)$ .



```
> cos(Pi/6);
tan(17*Pi/3);
ln(1);
```

$$\frac{\sqrt{3}}{2}$$

$$-\sqrt{3}$$

$$0$$

(6.9.1)

Maple simplifie donc automatiquement sous forme de radicaux les multiples entiers des arcs remarquables:

$$0, \frac{\pi}{6}, \frac{\pi}{4}, \frac{\pi}{3}, \frac{\pi}{2}, \pi$$

Pour exprimer le résultat avec des radicaux lorsque  $\pi$  est divisé par un diviseur de 120 ou de 24, il faudra faire une simplification sur demande en utilisant la macro-commande `convert` (voir `?convert,radical`).

```
> sin(5*Pi/12);
cos(7/60*Pi);
```

$$\sin\left(\frac{5\pi}{12}\right)$$

$$\cos\left(\frac{7\pi}{60}\right)$$

(6.9.2)

```
> sin(5*Pi/12)=convert(sin(5*Pi/12),radical);
cos(7/60*Pi)=convert(cos(7/60*Pi),radical);
```

$$\sin\left(\frac{5\pi}{12}\right) = \frac{\sqrt{2}(1+\sqrt{3})}{4}$$

$$\cos\left(\frac{7\pi}{60}\right) = \left(-\frac{1}{8} + \frac{\sqrt{3}}{8}\right) \sqrt{5-\sqrt{5}} + \frac{\sqrt{2}(\sqrt{5}+1)\sqrt{3}}{16} + \frac{\sqrt{2}(\sqrt{5}+1)}{16} \quad (6.9.3)$$

## Maple en tant que « métacalculatrice »

L'opérateur d'assignation « := » permet de donner un nom à des objets créés avec Maple. C'est le début de la sagesse... En effet, cette capacité de nommer les objets permet des développements qui, autrement, seraient laborieux et parfois même impossible. Mais cela rend nécessaire l'obligation de distinguer variable mathématique et variable informatique.

### Variable mathématique et variable informatique

Au cours d'une session Maple, il n'y a aucune déclaration à faire lorsqu'on utilise une variable mathématique. L'évaluateur la prend immédiatement en charge.

```
> x; # la variable x est libre (x est une variable mathématique)
```

 $x$ 

(7.1.1)

```
> x^2;
```

 $x^2$ 

(7.1.2)

```
> sin(theta+2*k*Pi);
```

$$\sin(2k\pi + \theta)$$

(7.1.3)

L'opérateur d'affectation ou d'assignation est le deux-points égal « := »

```
> x:=-5; # La variable x n'est plus libre ( x est maintenant une
variable informatique)
```

$$x := -5$$

(7.1.4)

```
> x^2;
```

$$25$$

(7.1.5)

Le résultat n'est pas  $x^2$  mais 25 car  $x$ , en tant que variable informatique, pointe vers la valeur 5.

Lorsqu'une variable est assignée, l'assignation demeure active tout au long de la séance de travail tant et aussi longtemps qu'aucune autre assignation ne viendra la modifier. C'est la responsabilité de l'utilisateur de gérer au moment de l'exécution d'une requête si les variables définissant les termes de la requête sont libres ou assignées.

Au cours d'une session Maple, on peut rendre à nouveau une variable mathématique à l'aide des apostrophes droites ([quotes](#)).

```
> x:='x';
```

$$x := x$$

(7.1.6)

Voyons voir.

```
> x^2;
```

$$x^2$$

(7.1.7)

Les apostrophes droites commandent à l'évaluateur de limiter l'évaluation au nom de la variable. Par contre, cette limitation ne limite en aucun temps le mécanisme de la simplification automatique en excluant la simplification des entrées de la table d'affectation système.

```
> sqrt(4^2+a^2+b+b+cos(0));
'sqrt(4^2+a^2+b+b+cos(0))';
```

$$\frac{\sqrt{a^2 + 2b + 17}}{\sqrt{16 + a^2 + 2b + \cos(0)}}$$

(7.1.8)

## Principe d'interactivité de Maple

Qu'est ce qu'une session Maple ? Lorsqu'on lance Maple, on lance un programme informatique où, dans un endroit bien précis de la feuille de travail, l'utilisateur inter-réagit avec ce programme. Cet endroit bien précis de la feuille de travail est la zone de requêtes, là où on trouve le bloc [ > ]. C'est par cette région de la feuille que l'utilisateur converse avec Maple. Ensuite, dans la zone des résultats (milieu de la feuille en bleu), l'utilisateur constate ce que Maple a « compris ». Toutes autres régions de la feuille de travail y compris la zone des résultats constituent seulement une trace écrite pour que l'utilisateur comprenne lui-même où il en est. Pour Maple, ces régions ne sont pas pris en compte.

Explorons ce principe d'interactivité. Exécuter la requête suivante et observer le résultat.

```
> x:=5;
```

Changer la lettre  $x$  par la lettre  $y$  dans la requête précédente et ne pas exécuter (ne pas presser la touche retour).

Exécuter ensuite la requête ci-dessous.

```
[ > x^2;
```

Qu'est-ce que vous constatez ? Cela vous semble-t-il curieux ? Quelle est votre explication ?

Dès l'instant où la requête  $x:=5;$  est exécutée, Maple fait pointer  $x$  vers la valeur 5. Après avoir changé la lettre  $x$  par la lettre  $y$  sans exécution de la requête,  $y$  demeure encore libre mais Maple pointe toujours  $x$  vers la valeur 5.

Exécuter maintenant la requête où vous avez changé la lettre  $x$  par la lettre  $y$  et exécuter de nouveau la même requête  $x^2$ .

Cela vous semble-t-il curieux ? Quelle est votre explication ?

Nous avons maintenant que  $y$  pointe vers la valeur 5 mais  $x$  également : Maple a... la mémoire longue.

Maintenant, dans la zone de requêtes immédiatement après le titre de cette section et avant le premier paragraphe, saisir dans ce bloc l'expression  $3x^2 - 5x + 1$  en terminant avec un point-virgule. Exécuter cette requête.

Qu'est-ce que vous constatez ? Le résultat vous semble-t-il encore ici curieux ? Quelle est votre explication ?

Comme  $x$  pointe vers la valeur 5, Maple a réduit cette expression par sa valeur, soit 51. Comme nous l'avons dit, l'utilisateur est en mode conversationnelle avec Maple et toutes zones de requêtes sont des endroits de la feuille de travail où Maple est à l'écoute. Toute autre endroit de la feuille de travail constitue une trace écrite pour l'utilisateur seulement.

Si vous n'avez pas fait correctement ce qui a été demandé, **avant de recommencer**, il faudra rendre libres les variables  $x$  et  $y$  en exécutant les requêtes ci-dessous.

Sinon, ne pas exécuter ces requêtes: il faut que la variable soit assignée avant de poursuivre.

```
[ > y:='y';  
  x:='x';
```

Créons un nouveau document Maple (**File->New** ou **Ctrl+N**) et exécuter la requête  $x^2 - 5$ . Qu'est-ce que vous avez encore constaté ? Le résultat vous semble-t-il encore ici curieux ? Quelle est votre explication ?

Le résultat devrait être  $x^2 - 5$  car, dans ce nouveau document, la variable  $x$  est libre à moins que l'on ait modifié dans l'onglet *Général* du sous-menu Options... la manière dont Maple devrait traiter la création de nouveaux moteurs mathématique. Si votre résultat est  $x^2 - 5$ , c'est qu'il y a eu la création d'un nouveau moteur. C'est habituellement cette configuration qui est réalisée lors de l'installation du logiciel Maple.

Fermer ce nouveau document sans le sauvegarder.

**Remarque:** Avant d'entreprendre tout autre développement dans une autre section, rendons libres les variables qui n'ont plus de raison d'être assignées.

```
[ > x:='x':  
  y:='y':
```

## La simplification sur demande

La macro-commande `radnormal` est la macro-commande dédiée à la simplification de nombres comportant des radicaux. Par exemple:

```
> Formule := (1-1/5*5^(1/2))*(2/(5^(1/2)-1))^12/(5^(1/2)-1)+(-1-1/5*5^(1/2))*(-2/(5^(1/2)+1))^12/(5^(1/2)+1);
```

$$\text{Formule} := \frac{4096 \left(1 - \frac{\sqrt{5}}{5}\right)}{(\sqrt{5} - 1)^{13}} + \frac{4096 \left(-1 - \frac{\sqrt{5}}{5}\right)}{(\sqrt{5} + 1)^{13}} \quad (9.1)$$

```
> radnormal(Formule);
```

$$144 \quad (9.2)$$

Il est même possible d'effectuer la rationalisation du dénominateur:

```
> Formule := (2+sqrt(6))/(sqrt(3)+sqrt(2));
```

$$\text{Formule} := \frac{2 + \sqrt{6}}{\sqrt{3} + \sqrt{2}} \quad (9.3)$$

```
> radnormal(Formule);
```

$$\frac{2 + \sqrt{3} \sqrt{2}}{\sqrt{3} + \sqrt{2}} \quad (9.4)$$

L'option `rationalized` permettra la rationalisation du dénominateur.

```
> radnormal(Formule,rationalized);
```

$$\sqrt{2} \quad (9.5)$$

Par simplification en mathématique, on entend habituellement réduction d'une expression. La macro-commande `simplify` est, en générale, la première macro-commande à laquelle on pense faire usage lorsqu'on veut simplifier une expression mais elle ne produit pas nécessairement le résultat attendu.

```
> Formule := (2+sqrt(6))/(sqrt(3)+sqrt(2));
```

$$\text{Formule} := \frac{2 + \sqrt{6}}{\sqrt{3} + \sqrt{2}} \quad (9.6)$$

```
> simplify(Formule);
```

$$\frac{2 + \sqrt{3} \sqrt{2}}{\sqrt{3} + \sqrt{2}} \quad (9.7)$$

Par simplification d'une expression, il faudrait plutôt entendre transformation de l'expression et cela, habituellement, vers une écriture plus appropriée à un contexte donné.

```
> Formule := 1 - (sin^2)(x);
```

$$\text{Formule} := 1 - \sin(x)^2 \quad (9.8)$$

```
> simplify(Formule);
```

$$\cos(x)^2 \quad (9.9)$$

Mais c'est pas toujours le cas avec `simplify`.

```
> Formule := sin(2*x)/cos(2*x);
```

$$\text{Formule} := \frac{\sin(2x)}{\cos(2x)} \quad (9.10)$$

```
> simplify(Formule);
```

$$\frac{\sin(2x)}{\cos(2x)} \quad (9.11)$$

Maple, de manière automatique ou sur demande n'estime pas que  $\tan(2x)$  est une écriture plus simple.

Utilisons alors la macro-commande `convert`.

```
> convert(Formule, tan);
```

$$\tan(2x) \quad (9.12)$$

La macro-commande `convert` est donc une macro-commande de simplification.

Mais, très souvent, `simplify` donne de bonne réduction.

```
> Formule := ((sin^3)(x) + (cos^3)(x)) / (sin(x) + cos(x));
```

$$Formule := \frac{\sin(x)^3 + \cos(x)^3}{\sin(x) + \cos(x)} \quad (9.13)$$

```
> simplify(Formule);
```

$$-\sin(x) \cos(x) + 1 \quad (9.14)$$

**ATTENTION:** Rappelons que c'est à vous qu'il revient de déterminer, au moment de la simplification (*automatique et/ou sur demande*), si la perte d'information est significative ou non.

Sans élaborer plus avant sur la notion de simplification sur demande, mentionnons que les principales macro-commandes disponibles pour réaliser des simplifications sur demande sont

[radnormal](#), [normal](#), [simplify](#),  
[expand](#), [factor](#), [combine](#),  
[collect](#), [convert](#), [rationalize](#),  
[sort](#), [subs](#), [algsubs](#).

## Il y a evalf et eval

Il ne faut pas confondre les macro-commandes `evalf` et `eval`. Rappelons que la macro-commande `evalf` est utile pour obtenir une approximation décimale d'une expression symbolique tandis que la macro-commande `eval` sert à évaluer une expression dont le résultat peut être exacte.

```
> Formule := sqrt(exp(x));
```

$$Formule := \sqrt{e^x} \quad (10.1)$$

```
> evalf(Formule, x=Pi);
```

*Error, invalid input: evalf expects its 2nd argument, n, to be of type posint, but received x = Pi*

Ce message d'erreur nous informe d'une erreur syntaxique dans la macro-commande `evalf`. En effet, il faut d'abord évaluer l'expression puis obtenir après seulement une approximation de cette évaluation.

```
> eval(Formule, x=Pi);
```

$$\sqrt{e^\pi} \quad (10.2)$$

```
> evalf(eval(Formule, x=Pi));
```

$$4.810477382 \quad (10.3)$$

La dernière requête a permis d'obtenir une approximation de la valeur exacte de `eval (Formule , x=Pi )`, c'est-à-dire de  $\sqrt{e^\pi}$ .

De plus, sur un autre plan, il est utile d'utiliser la macro-commande `eval` pour bien documenter la zone des résultats.

```
> eval(Formule,x=Pi)=evalf(eval(Formule,x=Pi));
```

$$\sqrt{e^\pi} = 4.810477382 \quad (10.4)$$

Voici un autre exemple illustrant une bonne documentation de la zone des résultats. Après avoir créé la variable `Volume_du_parallélépipède` pour le calcul du volume d'un parallélépipède rectangle comme suit:

```
> Volume_du_parallélépipède:=Longueur*Largeur*Hauteur;
```

$$\text{Volume\_du\_parallélépipède} := \text{Longueur} \text{Largeur} \text{Hauteur} \quad (10.5)$$

Précédons ensuite à l'évaluation de la formule avec les dimensions suivantes. Remarquer l'utilisation des crochets.

```
> eval(Volume_du_parallélépipède,[Longueur=4,Largeur=3,Hauteur=8]);
```

$$96 \quad (10.6)$$

Utilisons des apostrophes droites pour bien documenter la zone des résultats en formulant le résultat sous la forme d'une égalité.

```
> 'Volume_du_parallélépipède'=eval(Volume_du_parallélépipède,[Longueur=4,Largeur=3,Hauteur=8]);
```

$$\text{Volume\_du\_parallélépipède} = 96 \quad (10.7)$$

L'utilisation des apostrophes droites autour de la variable informatique `Volume` est nécessaire afin de ne pas afficher ce vers quoi elle pointe.

```
> Volume_du_parallélépipède=eval(Volume_du_parallélépipède,[Longueur=4,Largeur=3,Hauteur=8]);
```

$$\text{Longueur} \text{Largeur} \text{Hauteur} = 96 \quad (10.8)$$

## Il y a variable et variable

Il ne faut pas perdre de vue, au moment de la création ou de l'utilisation d'une formule, si les variables utilisées sont libres (variables mathématiques) ou si elles sont assignées (variables informatiques) car cela peut mener à des erreurs «logiques».

```
> x:=2;
```

$$x := 2 \quad (11.1)$$

```
> eval(x/(x+1),[x=100]); # Ici les crochets sont facultatifs
```

$$\frac{2}{3} \quad (11.2)$$

Compte tenu de l'état du système au moment de la requête d'évaluation, Maple a "évalué" la constante  $\frac{2}{3}$ .

Maple n'a pas pu évaluer la formule  $\frac{x}{x+1}$  avec  $x = 100$  comme on aurait voulu qu'il le fasse puisque, par l'assignation  $x=2$ , la variable  $x$  était informatique. Ainsi, la variable  $x$  pointant vers la valeur 2, la formule

$\frac{x}{x+1}$  pointe alors vers le nombre  $\frac{2}{3}$  au moment de l'évaluation avec la macro-commande `eval`.

Pour effectuer l'évaluation attendue, il faut rendre d'abord la variable  $x$  mathématique, c'est-à-dire libre.

```
> x:='x';
```

$$x := x \quad (11.3)$$

Ensuite, l'évaluation pourra être faite correctement.

```
> eval(x/(x+1), [x=100]);
```

$$\frac{100}{101} \quad (11.4)$$

## La macro-commande `restart`

La macro-commande « `restart` » de la bibliothèque de base est utilisée pour remettre l'environnement de la séance de travail dans l'état où il était au moment du lancement de Maple: tout est oublié. Après un redémarrage, la quantité de mémoire allouée par le noyau Maple devrait être similaire à la quantité de mémoire allouée lors du premier démarrage d'un noyau Maple. Cette macro-commande est habituellement utilisée une seule fois dans un même document Maple, soit seulement au tout début de la lecture d'une nouvelle feuille Maple. En effet, en exécutant la requête `restart` dans n'importe quelle zone de requête, on désassigne bien sûr toutes les variables mais également toutes les indications précisant les emplacements des macro-commandes spécifiées avec la macro-commande `with` et on remet toutes les variables systèmes à leur valeur par défaut. Ce qui, généralement, n'est pas toujours désiré.

Alors, pour désassigner certaines variables seulement sans modifier l'état courant de la session de travail (variables systèmes et macro-commandes spécifiées avec `with`), il vaut donc mieux le faire comme suit à l'aide des apostrophes droites:

```
> x:='x':  
toto:='toto':
```

Il est possible aussi d'utiliser la macro-commande `unassign` pour désassigner simultanément plusieurs variables.

```
> unassign('x,toto,a,b,c');
```

## Variable d'environnement `Digits` et chiffres significatifs

La variable `Digits` est une variable d'environnement qui impose à Maple le nombre de chiffres qui sera utilisé dans les approximations demandées. Sa valeur par défaut est 10 mais l'utilisateur peut, en tout temps, lui assigner le nombre voulu de chiffres.

Une remarque s'impose immédiatement: **il ne faut pas confondre nombres de chiffres utilisés dans les calculs et nombre de chiffres significatifs du résultat**. Explorons le calcul du nombre

$$\sin\left(\frac{\pi}{5}\right) \left( e^{\frac{2+\sqrt{6}}{\sqrt{2}+\sqrt{3}}} \right)^{\sqrt{2}}$$

lorsque la variable `Digits` est initialisé avec la valeur 10 (valeur par défaut). Dans un premier temps nous allons obtenir une approximation avec les calculs fait avec 10 chiffres.

```
> Nombre:=sin(Pi/5)*exp((2+sqrt(6))/(sqrt(2)+sqrt(3)))^sqrt(2);
```

$$\text{Nombre} := \sin\left(\frac{\pi}{5}\right) \left( e^{\frac{2+\sqrt{6}}{\sqrt{3}+\sqrt{2}}} \right)^{\sqrt{2}} \quad (13.1)$$

```
> evalf(Nombre);
```

$$4.343178199 \quad (13.2)$$

On peut penser qu'avec les cumuls d'erreurs suite aux arrondissements des calculs intermédiaires, Maple donne quant même le résultat avec 10 chiffres exacts ou peut-être plutôt avec 9.

Obtenons une seconde approximation de ce nombre mais, cette fois, en simplifiant ce nombre afin d'avoir moins d'opérations de calcul définissant ce nombre.

```
> Nombre_simplifié:=combine(simplify(convert(radnormal(Nombre),radical)
),radical);
```

$$\text{Nombre\_simplifié} := \frac{e^2 \sqrt{10 - 2\sqrt{5}}}{4} \quad (13.3)$$

Évidemment, la requête précédente qui imbrique plusieurs macro-commandes résume plusieurs simplifications successives qui ont été préalablement obtenues avec un développement pas à pas.

Obtenons maintenant cette seconde approximation.

```
> Nombre=evalf(Nombre);
Nombre_simplifié=evalf(Nombre_simplifié);
```

$$\sin\left(\frac{\pi}{5}\right) \left( e^{\frac{2+\sqrt{6}}{\sqrt{3}+\sqrt{2}}} \right)^{\sqrt{2}} = 4.343178199$$

$$\frac{e^2 \sqrt{10 - 2\sqrt{5}}}{4} = 4.343178202 \quad (13.4)$$

On constate alors qu'avec `Digits=10`, on aurait en fait 7 chiffres significatifs dans la première approximation *Nombre*.

En résumé, la valeur de la variable `Digits` ne signifie donc pas le nombre de chiffres significatifs désirés dans le résultat mais seulement que cette valeur indique à Maple d'utiliser un nombre donné de chiffres dans ses évaluations numériques.

Sans autres explications, recalculons en augmentant la variable d'environnement `Digits` à 12.

```
> Digits:=12;
```

$$\text{Digits} := 12 \quad (13.5)$$

```
> Nombre=evalf(Nombre);
Nombre_simplifié=evalf(Nombre_simplifié);
```

$$\sin\left(\frac{\pi}{5}\right) \left( e^{\frac{2+\sqrt{6}}{\sqrt{3}+\sqrt{2}}} \right)^{\sqrt{2}} = 4.34317820328$$



$$\frac{e^2 \sqrt{10 - 2\sqrt{5}}}{4} = 4.34317820332 \quad (13.6)$$

```
> Digits:=15;
  Nombre:=evalf(Nombre);
  Nombre_simplifié:=evalf(Nombre_simplifié);
```

$$\begin{aligned} & \text{Digits} := 15 \\ \sin\left(\frac{\pi}{5}\right) \left(e^{\frac{2+\sqrt{6}}{\sqrt{3}+\sqrt{2}}}\right)^{\sqrt{2}} &= 4.34317820331319 \end{aligned}$$

$$\frac{e^2 \sqrt{10 - 2\sqrt{5}}}{4} = 4.34317820331318 \quad (13.7)$$

## Le référentiel de Maple

Le mécanisme de la simplification automatique de Maple compose avec l'ensemble des [nombres complexes](#). Observons ce mécanisme lors de l'extraction de racines avec les macro-commandes [root](#) et [surd](#).

*Notons, que pour l'évaluateur, les expressions créées avec des puissances fractionnaires ou avec la macro-commande **root** sont équivalentes.*

```
> r:=(-8)^(1/3);
```

$$r := (-8)^{1/3} \quad (14.1)$$

```
> r:=simplify((-8)^(1/3));
```

$$r := 1 + I\sqrt{3} \quad (14.2)$$

Est-ce que ce résultat vous surprend ? Montrons que c'est bien une racine cubique de -8 en élevant au cube ce résultat.

```
> r^3;
```

$$(1 + I\sqrt{3})^3 \quad (14.3)$$

Obtenons avec [evalc](#) l'écriture algébrique  $a + bi$  du nombre  $(1 + \sqrt{3} I)^3$ .

```
> evalc(r^3);
```

$$-8 \quad (14.4)$$

Ce qui montre que **r** est bien un nombre dont le cube est -8.

```
> r:=root(-8,3);
```

$$r := 2(-1)^{1/3} \quad (14.5)$$

```
> simplify(r);
```

$$1 + I\sqrt{3} \quad (14.6)$$

Dans un cas comme dans l'autre,  $(-8)^{\frac{1}{3}}$  ou  $\text{root}(-8,3)$ , le nombre **r** ainsi défini est donc le même nombre complexe.

L'utilisation de la macro-commande **surd** est nécessaire pour extraire une racines  $n$ -ième dans les nombres réels (lorsqu'il y en a).

```
[ > surd(-8,3);
```

$$-2 \qquad (14.7)$$

Pour aller un peu plus loin avec les nombres complexes, consultez sur mon site Internet la feuille [Les nombres complexes avec Maple](#).

```
[ > r:='r':
```